Ideal intersection examples.

The example in the text on page 186 where $I = \langle x^2\, y\rangle$ and $J = \langle x\, y^2\rangle$ then $I \cap J = \langle x^2\, y^2\rangle$.

```
> f := x^2*y;
  g := x*y^2;
```

$$f := x^2\, y$$

$$g := x\, y^2$$

The algorithm is to eliminate t from the ideal $\langle f\, t,\, (1-t)\, g\rangle$.

```
> G := Groebner[Basis]( [f*t, (1-t)*g], plex( t,x,y ) );
```

$$G := [y^2\, x^2,\, t\, x\, y^2 - x\, y^2,\, x^2\, y\, t]$$

```
> remove(has,G,t);
```

$$[y^2\, x^2]$$

Consider $I = \langle x,\, y\rangle$ and $J = \langle x,\, z\rangle$. Again, clearly $I \cap J = \langle x,\, x\, z,\, y\, x,\, y\, z\rangle = \langle x,\, y\, z\rangle$

```
> G := Groebner[Basis]( [x*t, y*t, (1-t)*x, (1-t)*z], plex(t,x,y,z) );
  remove(has,G,t);
```

$$G := [z\, y,\, x,\, t\, z - z,\, y\, t]$$

$$[z\, y,\, x]$$

You can also use the PolynomialIdeals package which does this for you automatically.

```
> with(PolynomialIdeals):
> J := Intersect( <x,y>, <x,z> );
```

$$J := \langle x,\, z\, y\rangle$$

The second example in the text on page 186 is also simple because each ideal is a principal ideal except that the Groebner basis is quite big so I have not printed it.

```
> f := (x+y)^4*(x^2+y)^2*(x-5*y);
  g := (x+y)*(x^2+y)^3*(x+3*y);
```

$$f := (x+y)^4\, (x^2+y)^2\, (x-5\, y)$$

$$g := (x+y)\, (x^2+y)^3\, (x+3\, y)$$

```
> G := Groebner[Basis]([f*t,(1-t)*g], plex(t,x,y) ):
  E := remove(has,G,t);
```

$$E := [x^{12} + 2\, x^{11}\, y - 17\, x^{10}\, y^2 - 68\, x^9\, y^3 - 97\, x^8\, y^4 - 62\, x^7\, y^5 - 15\, x^6\, y^6 + 3\, x^{10}\, y + 6\, x^9\, y^2 - 51\, x^8\, y^3 - 204\, x^7\, y^4 - 291\, x^6\, y^5 - 186\, x^5\, y^6 - 45\, x^4\, y^7 + 3\, x^8\, y^2 + 6\, x^7\, y^3 - 51\, x^6\, y^4 - 204\, x^5\, y^5 - 291\, x^4\, y^6 - 186\, x^3\, y^7 - 45\, x^2\, y^8 + x^6\, y^3 + 2\, x^5\, y^4 - 17\, x^4\, y^5 - 68\, x^3\, y^6 - 97\, x^2\, y^7 - 62\, x\, y^8 - 15\, y^9]$$

```
> h := factor(E[1]);
```

$$h := (x+3\, y)\, (x-5\, y)\, (x^2+y)^3\, (x+y)^4$$

This is the LCM(f, g) so GCD(f, g) is

```
> simplify( f*g/h );
```

$$(x+y)\, (x^2+y)^2$$